*******************************************************************************
*** Description: 	This document provides the code for reproducing the 	***
***					figures in Study 2 of the paper, "Compulsory Voting 	***
***					Diminishes the Relationship Between Winning and 		***
***					Satisfaction with Democracy," which is authored by 		***
***					Shane P. Singh and appears in the Journal of Politics	***
***					 														***
***					It also provides the code for reproducing the tables 	***
***					and figures associated with Study 2 in the 				***
***					Supplementary Material.									***
***					 														***
***					It also provides the code for reproducing statistics 	***
***					associated with claims related to Study 2 made in the  	***
***					text.													***
*******************************************************************************


**************
**************
*Set the Version                                                                                                                                 
**************
**************
version 17.0


**************
**************
*Install Required Package (remove initial asterisk if package not yet installed)                                                                                                                     
**************
**************
*ssc install coefplot 


**************
**************
*Open the Required Dataset
**************
**************
use "CSES_IMD_JOP_Replication.dta", clear


**************
**************
*Create a Variable that Identifies the Estimation Sample
**************
**************
reg satdem i.voted_winner_PM_PREZ   educ_scale_0_10  female age ideo majoritarian dem_development latin if ageCV~=0, cl(cntryyear)
gen samp = 1 if e(sample)



**************
**************
*Evidence for Claim Made in the Text: "About 17 percent of respondents in my CSES sample of 71,654 individuals are subject to enforced compulsory voting."
**************
**************	
tab  comp_enforced_VoD_v11  if samp==1



**************
**************
*Evidence for Claim Made in the Text: "The average offcial turnout rate is 14.8 percentage points higher in the elections conducted with enforced compulsory voting than otherwise (two-sided p < 0.001)."
**************
**************	
preserve
collapse  v2elvaptrn  comp_enforced_VoD_v11  if samp == 1, by(cntryyear)
reg v2elvaptrn  comp_enforced_VoD_v11, 
restore


**************
**************
*Figure 5
**************
**************
meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 
estimates store Study_2_meologit_main

margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_main_xb

coefplot ///
		(, offset(0) recast(scatter) mcolor(black) msize(medsmall) ciopts(lpattern(solid) lcolor(black))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Latent Satisfaction with Democracy" " ") ///
	legend(off) 	///			 
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(small)) ///
	ylabel(.4(0.1).7)  yscale(range(.37 .73)) ///
	xlabel(, labsize(small)) ///
	yline(0,   lcolor(gs3) lwidth(thin) lpattern(dash)) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(4.8)  ysize(7) scale(1.31) level(90) format(%9.1f)

	

**************
**************
*Evidence for Claim Made in the Text: "... the two-sided p-value of the difference in the impact of winning on satisfaction across elections with and without enforced compulsory voting is 0.012."
**************
**************		
estimates restore Study_2_margins_main_xb
margins, coeflegend
test  _b[1.voted_winner_PM_PREZ:1bn._at] = _b[1.voted_winner_PM_PREZ:2._at] //*same as p-value on the interaction between winning and enforced compulsory voting


**************
**************
*Evidence for Claim Made in the Text: "Results are substantively unchanged if I code winners in parliamentary systems as those who voted for any party that entered government."
**************
**************	
meologit satdem i.voted_winner_GOV_PREZ##i.comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 


**************
**************
*Evidence for Claim Made in the Text: "Substantive results do not change with the inclusion of an additional control that directly differentiates parliamentary and presidential elections."
**************
**************	
meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin parliamentary_election if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 



**************
**************
*Figure 6
**************
**************
*first category
estimates restore 	Study_2_meologit_main 
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(outcome(1) fixedonly) post level(90) 
estimates store Study_2_margins_main_cat_1

*second category
estimates restore 	Study_2_meologit_main
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(outcome(2) fixedonly) post level(90) 
estimates store Study_2_margins_main_cat_2

*third category
estimates restore 	Study_2_meologit_main
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(outcome(3) fixedonly) post level(90) 
estimates store Study_2_margins_main_cat_3

*fourth category
estimates restore 	Study_2_meologit_main
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(outcome(4) fixedonly) post level(90) 
estimates store Study_2_margins_main_cat_4

coefplot ///
		(Study_2_margins_main_cat_1, offset(-.15) recast(scatter) mcolor(black) msize(medsmall) ciopts(lpattern(solid) lcolor(black))) ///
		(Study_2_margins_main_cat_2, offset(-.05) recast(scatter) mcolor(red) msize(medsmall) ciopts(lpattern(solid) lcolor(red))) ///
		(Study_2_margins_main_cat_3, offset( .05) recast(scatter) mcolor(blue) msize(medsmall) ciopts(lpattern(solid) lcolor(blue))) ///
		(Study_2_margins_main_cat_4, offset( .15) recast(scatter) mcolor(green) msize(medsmall) ciopts(lpattern(solid) lcolor(green))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Pr(in Satisfaction Category)" " ") ///
	legend(order(						///
			1 "Not at all Satisfied"  	///
			3 "Not Very Satisfied" 		///
			5 "Fairly Satisfied"		///
			7 "Very Satisfied") 		///
				rows(2) size(vsmall)) 	///			 
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(small)) ///
	ylabel(-.1(0.1)0.1)  yscale(range(-.11 .11)) ///
	xlabel(, labsize(small)) ///
	yline(0,   lcolor(gs3) lwidth(thin) lpattern(dash)) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(6.2)  ysize(8) scale(1.05) level(90) format(%9.1f)

	

**************
**************
*Evidence for Claim Made in the Text: "... a winner who was not compelled to vote is about 9 percentage points more likely to report being 'fairly satisfied' than a non-compelled voter who selected a losing party or candidate. Among those voting under enforced compulsory rules, the impact of winning on the probability of being 'fairly satisfied' is less than half the size. The difference between the associations, while substantively meaningful, misses conventional levels of statistical significance (two-sided p = 0.141)."
**************
**************	
estimate restore Study_2_margins_main_cat_3
margins, coeflegend 
test _b[1.voted_winner_PM_PREZ:1bn._at] =  _b[1.voted_winner_PM_PREZ:2._at]	
	
	
	
**************
**************
*Evidence for Claim Made in the Text: "... the impact of winning on being 'not very satisfied' where abstention is not sanctioned is almost -10 percentage points. Where nonvoting attracts a penalty, it is just -7 percentage points. The difference in these associations has a two-sided p-value of 0.024."
**************
**************	
estimate restore Study_2_margins_main_cat_2
margins, coeflegend 
test _b[1.voted_winner_PM_PREZ:1bn._at] =  _b[1.voted_winner_PM_PREZ:2._at]	
	
	
	
**************
**************
*Figure 7
**************
**************	
***create a scale from a set of covariates that affect turnout
*transform age by subtracting each individual's age in years from 55, squaring this difference, and reversing its sign
gen age_sq_dist_55_neg = .
replace age_sq_dist_55_neg = -((age-55)^2)

* create the volitional voting scale by standardizing each component and summing within respondents
alpha educ_scale_0_10  income   age_sq_dist_55_neg if samp == 1, std  gen(turnout_scale) asis
sum turnout_scale if samp == 1, detail
global p10 = r(p10)
global med = r(p50)
global p90 = r(p90)


meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11##c.turnout_scale female ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 
estimates store Study_2_meologit_3way_vol


estimates restore Study_2_meologit_3way_vol
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) turnout_scale = ($p10)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_10

estimates restore Study_2_meologit_3way_vol
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) turnout_scale = ($med)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_50

estimates restore Study_2_meologit_3way_vol
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) turnout_scale = ($p90)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_90

coefplot ///
		(Study_2_margins_3way_xb_10, offset(-.2) recast(connected) mcolor(black) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(black) ciopts(lpattern(solid) lcolor(black))) ///
		(Study_2_margins_3way_xb_50, offset(0) recast(connected) mcolor(red) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(red) ciopts(lpattern(solid) lcolor(red))) ///
		(Study_2_margins_3way_xb_90, offset(.2) recast(connected) mcolor(blue) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(blue) ciopts(lpattern(solid) lcolor(blue))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Latent Satisfaction with Democracy" " ") ///
		legend(order(						///
			1 "Low Turnout Volition"  	///
			3 "Median Turnout Volition" 		///
			5 "High Turnout Volition")		///
				rows(3) size(vsmall)) 	///		
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(small)) ///
	ylabel(.4(0.1).7)  yscale(range(.37 .73)) ///
	xlabel(, labsize(small) ) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(4.8)  ysize(7) scale(1.31) level(90) format(%9.1f)


**************
**************
*Evidence for Claim Made in the Text: "For those at the 10th percentile of turnout volition, the difference in the association between winning and latent satisfaction across elections with and without enforced compulsory voting is 0.19 (in terms of the ordered log-odds). For those at the 90th percentile, the difference is essentially nonexistent. The difference in these differences is statistically distinguishable from zero (two-sided p = 0.081)."
**************
**************	
estimates restore Study_2_meologit_3way_vol
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) turnout_scale = ($p10 $med $p90)) predict(xb fixedonly) post level(90) coeflegend
lincom _b[1.voted_winner_PM_PREZ:1bn._at] - _b[1.voted_winner_PM_PREZ:4._at] //*difference in the effect of winning for tenth percentile of turnout volition respondents
lincom _b[1.voted_winner_PM_PREZ:3._at] - _b[1.voted_winner_PM_PREZ:6._at] //*difference in the effect of winning for ninetieth percentile of turnout volition respondents
test   (_b[1.voted_winner_PM_PREZ:1bn._at] - _b[1.voted_winner_PM_PREZ:4._at]) = (_b[1.voted_winner_PM_PREZ:3._at] - _b[1.voted_winner_PM_PREZ:6._at]) //*difference in the differences in the effect of winning for tenth percentile and ninetieth percentile of turnout volition. *same as p-value on the interaction between winning, enforced compulsory voting, and turnout volition



**************
**************
*Evidence for Claim Made in the Text: "Results are robust to a logarithmic transformation."
**************
**************	
gen age_folded_55 = age
replace age_folded_55 = -age + 110 if age > 55
gen age_ln_dist_55 = .
replace age_ln_dist_55 = ln(age_folded_55) if samp == 1
alpha educ_scale_0_10  income   age_ln_dist_55 if samp == 1, std  gen(turnout_scale_ln) asis

meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11##c.turnout_scale_ln female ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 


**************
**************
*Table SM3
**************
**************
estimates restore Study_2_meologit_main
meologit


**************
**************
*Table SM4
**************
**************
estimates restore Study_2_meologit_3way_vol
meologit


**************
**************
*Table SM6
**************
**************
table countryandyear_no_underscore comp_enforced_VoD_v11 if samp == 1



**************
**************
*Figure SM11
**************
**************	
mixed satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, mle 
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1)) predict(xb) post level(90) 
estimates store Study_2_margins_linear_xb

coefplot ///
		(, offset(0) recast(scatter) mcolor(black) msize(medsmall) ciopts(lpattern(solid) lcolor(black))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Satisfaction with Democracy" " ") ///
	legend(off) 	///			 
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(medsmall)) ///
	ylabel(.1(0.05).3)   ///
	xlabel(, labsize(small)) ///
	yline(0,   lcolor(gs3) lwidth(thin) lpattern(dash)) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(4.8)  ysize(7) scale(1.31) level(90) format(%9.2f)


**************
**************
*Evidence for Claim Made in the Text: "... the difference in the impact of winning on satisfaction across elections with and without enforced compulsory voting has a two-sided p-value of 0.017."
**************
**************	
estimates restore Study_2_margins_linear_xb
margins, coeflegend
test  _b[1.voted_winner_PM_PREZ:1bn._at] = _b[1.voted_winner_PM_PREZ:2._at] //*same as p-value on the interaction between winning and enforced compulsory voting



**************
**************
*Figure SM12
**************
**************	
*create external efficacy scale
alpha matters_who_vote_for  matters_who_is_in_power, gen(external_efficacy) asis casewise
sum external_efficacy if samp == 1, detail
global p10 = r(p10)
global med = r(p50)
global p90 = r(p90)

meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11##c.external_efficacy  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 
estimates store Study_2_meologit_3way_eff


estimates restore Study_2_meologit_3way_eff
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) external_efficacy = ($p10)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_10

estimates restore Study_2_meologit_3way_eff
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) external_efficacy = ($med)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_50

estimates restore Study_2_meologit_3way_eff
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) external_efficacy = ($p90)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_90

coefplot ///
		(Study_2_margins_3way_xb_10, offset(-.2) recast(connected) mcolor(black) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(black) ciopts(lpattern(solid) lcolor(black))) ///
		(Study_2_margins_3way_xb_50, offset(0) recast(connected) mcolor(red) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(red) ciopts(lpattern(solid) lcolor(red))) ///
		(Study_2_margins_3way_xb_90, offset(.2) recast(connected) mcolor(blue) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(blue) ciopts(lpattern(solid) lcolor(blue))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Latent Satisfaction with Democracy" " ") ///
		legend(order(						///
			1 "Low Efficacy"  	///
			3 "Median Efficacy" 		///
			5 "High Efficacy")		///
				rows(3) size(vsmall)) 	///		
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(small)) ///
	ylabel(.1(0.2).9)  yscale(range(.1 .9)) ///
	xlabel(, labsize(small) ) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(4.8)  ysize(7) scale(1.31) level(90) format(%9.1f)

	
**************
**************
*Evidence for Claim Made in the Text: "For those at the 10th percentile of efficacy, the difference in the association between winning and latent satisfaction across elections with and without enforced compulsory voting is 0.21 (in terms of the ordered log-odds). For those at the 90th percentile of external efficacy, the difference is about half the size, at 0.10."
**************
**************	
estimates restore Study_2_meologit_3way_eff
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) external_efficacy = ($p10 $med $p90)) predict(xb fixedonly) post level(90) 
lincom _b[1.voted_winner_PM_PREZ:1bn._at] - _b[1.voted_winner_PM_PREZ:4._at] //*difference in the effect of winning for tenth percentile of efficacy 
lincom _b[1.voted_winner_PM_PREZ:3._at] - _b[1.voted_winner_PM_PREZ:6._at] //*difference in the effect of winning for ninetieth percentile of efficacy 

	
drop external_efficacy


**************
**************
*Figure SM13
**************
**************	
sum mean_dev_from_LR_ind if samp == 1, detail
global p10 = r(p10)
global med = r(p50)
global p90 = r(p90)

meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11##c.mean_dev_from_LR_ind  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 
estimates store Study_2_meologit_3way_acc

estimates restore Study_2_meologit_3way_acc
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) mean_dev_from_LR_ind = ($p10)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_10

estimates restore Study_2_meologit_3way_acc
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) mean_dev_from_LR_ind = ($med)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_50

estimates restore Study_2_meologit_3way_acc
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) mean_dev_from_LR_ind = ($p90)) predict(xb fixedonly) post level(90) 
estimates store Study_2_margins_3way_xb_90

coefplot ///
		(Study_2_margins_3way_xb_10, offset(-.2) recast(connected) mcolor(black) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(black) ciopts(lpattern(solid) lcolor(black))) ///
		(Study_2_margins_3way_xb_50, offset(0) recast(connected) mcolor(red) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(red) ciopts(lpattern(solid) lcolor(red))) ///
		(Study_2_margins_3way_xb_90, offset(.2) recast(connected) mcolor(blue) msize(medsmall) lwidth(vthin) lpattern(dash) lcolor(blue) ciopts(lpattern(solid) lcolor(blue))) ///
	,  scheme(s1color) vertical ///
	ytitle("Association Between Winning" "and Latent Satisfaction with Democracy" " ") ///
		legend(order(						///
			1 "Low Perceptual Error"  	///
			3 "Median Perceptual Error" 		///
			5 "High Perceptual Error")		///
				rows(3) size(vsmall)) 	///		
	coeflabels( 1._at  = 	`" "No Enforced" "Compulsory Voting" "'  ///
				2._at  = 	`" "Enforced" "Compulsory Voting" "' ) ///
	graphregion(margin(small)) ///
	ylabel(.1(0.2).9)  yscale(range(.1 .9)) ///
	xlabel(, labsize(small) ) ///
	xline(1.5, lcolor(gs3) lwidth(medium) lpattern(solid)) ///
	xsize(4.8)  ysize(7) scale(1.31) level(90) format(%9.1f)


	
**************
**************
*Evidence for Claim Made in the Text: "For those at the 90th percentile of perceptual error, the difference in the association between winning and latent satisfaction across elections with and without enforced compulsory voting is 0.17 (in terms of the ordered log-odds). For those at the 10th percentile of perceptual error, the difference is  0.14."
**************
**************	
estimates restore Study_2_meologit_3way_acc
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) mean_dev_from_LR_ind = ($p10 $med $p90)) predict(xb fixedonly) post level(90) 
lincom _b[1.voted_winner_PM_PREZ:1bn._at] - _b[1.voted_winner_PM_PREZ:4._at] //*difference in the effect of winning for tenth percentile of perceptual error  
lincom _b[1.voted_winner_PM_PREZ:3._at] - _b[1.voted_winner_PM_PREZ:6._at] //*difference in the effect of winning for ninetieth percentile of perceptual error  



**************
**************
*Evidence for Claim Made in the Text: "Substantive conclusions are unchanged when I use these expert placements rather than the averaged individual-level placements."
**************
**************	
sum mean_dev_from_LR_exp if samp == 1, detail
global p10 = r(p10)
global med = r(p50)
global p90 = r(p90)

meologit satdem i.voted_winner_PM_PREZ##i.comp_enforced_VoD_v11##c.mean_dev_from_LR_exp  educ_scale_0_10   female age ideo  majoritarian dem_development latin if samp ==1 || cntryyearnum: ideo, intmethod(mcaghermite) 
margins, dydx(voted_winner_PM_PREZ) at(comp_enforced_VoD_v11 = (0 1) mean_dev_from_LR_exp = ($p10 $med $p90)) predict(xb fixedonly) post level(90) 
lincom _b[1.voted_winner_PM_PREZ:1bn._at] - _b[1.voted_winner_PM_PREZ:4._at] //*difference in the effect of winning for tenth percentile of perceptual error (expert placements) 
lincom _b[1.voted_winner_PM_PREZ:3._at] - _b[1.voted_winner_PM_PREZ:6._at] //*difference in the effect of winning for ninetieth percentile of perceptual error (expert placements)  




**************
**************
*I estimated interactive multilevel ordered logit mediation models in which I allowed my measures of policy concern and psycho-emotional investment to intervene in the pathway from compulsory voting to the impact of winning on satisfaction. With regard to policy concern, I find the direct effect of compulsory voting to be 0.35 (two-sided p = 0.167), which is less than five percent of the range of the perceptual deviation variable used to capture policy concern. With regard to psycho-emotional investment, I find the direct effect of compulsory voting to be -0.02 (two-sided p = 0.927), which is equivalent to about one half of one percent of the range of the efficacy variable used to capture psycho-emotional investment.
**************
**************	
gsem (satdem <- i.voted_winner_PM_PREZ##c.mean_dev_from_LR_ind educ_scale_0_10   female age ideo  majoritarian dem_development latin M1[cntryyearnum] c.ideo#M2[cntryyearnum], ologit)  (mean_dev_from_LR_ind <-comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin  M3[cntryyearnum]) if samp ==1 , intmethod(mcaghermite) intpoints(3)  vce(opg)

sum mean_dev_from_LR_ind if samp == 1
global range = r(max) - r(min)
display .3494326/$range

*create external efficacy scale
alpha matters_who_vote_for  matters_who_is_in_power, gen(external_efficacy) asis casewise

gsem (satdem <- i.voted_winner_PM_PREZ##c.external_efficacy educ_scale_0_10   female age ideo  majoritarian dem_development latin M1[cntryyearnum] c.ideo#M2[cntryyearnum], ologit)  (external_efficacy <-comp_enforced_VoD_v11  educ_scale_0_10   female age ideo  majoritarian dem_development latin  M3[cntryyearnum]) if samp ==1 , intmethod(mcaghermite) intpoints(3)   vce(opg)

sum external_efficacy if samp == 1
global range = r(max) - r(min)
display -.0175741/$range




***           ___--=--------___
***          /. \___\____   _, \_      
***         /. .  _______     __/=====@		Thanks for replicating!
***         \----/  |  / \______/      
***     _/         _/ o \
***    / |    o   /  ___ \
***   / /    o\\ |  / O \ /|      __-_
***  |o|     o\\\   |  \ \ /__--o/o___-_
***  | |      \\\-_  \____  ----  o___-
***  |o|       \_ \     /\______-o\_-
***  | \       _\ \  _/ / |
***  \o \_   _/      __/ /
***   \   \-/   _       /|_
***    \_      / |   - \  |\
***      \____/  \ | /  \   |\
***              | o |   | \ |
***              | | |    \ | \
***             / | /      \ \ \
***           /|  \o|\--\  /  o |\--\
***           \----------' \---------'
